Subject: Re: BDHS2014-bdkr data set- Children nutrition- table 11.6 in BDHS report, 4+food group
Posted by Bridgette-DHS on Tue, 30 Aug 2016 14:56:15 GMT
View Forum Message <> Reply to Message

Following is a response from DHS Senior Research Associate, Shireen Assaf:

To match the figure in table 11.6 in the BDHS report of 2442 children with 27.6% having consumed 4+ food groups, you need to select for the youngest child living with the mother between the age of 6 and 23 months. This is not so straight forward as you will need to find the youngest child for each mother. The below code will allow you to do this.

```
*** Find the youngest child living with mother age 6-23 months ******
use BDKR70FL.DTA, clear
*keep only children that are living with the mother
keep if b9==0
sort for the merge later
sort v001 v002 v003 bidx
save temp.dta, replace

*finding the youngest child for each mother
collapse (min) bidx, by(v001 v002 v003)
sort v001 v002 v003 bidx
merge v001 v002 v003 bidx using temp.dta

gen ageC=v008-b3
keep if ageC>=6 & ageC<=23

keep if _merge==3
drop _merge

*** creating the variable diet4 (4 +food groups)

* food groups
gen grains=1 if  v412a==1| v414e==1|v414f==1
label value grains grains
label variable grains "GRAINS, ROOTS and TUBER"
tab grains

gen lagumes=1 if v414o==1
label value lagumes lagumes
label variable lagumes "LEGUMES and NUTS"

gen dairyp=1 if v411==1 | v411a==1 | v414p==1 |v414v==1
label value dairyp dairyp
label variable dairyp "DAIRY PRODUCTS"

gen fleshf=1 if v414h==1 | v414m==1 | v414n==1
```

```
label value fleshf fleshf
label variable fleshf "FLESH FOODS"

gen eggs=1 if v414g==1
label value eggs eggs
label variable eggs "EGGS"

gen vitama=1 if v414i==1|v414j==1|v414k==1
label value vitama vitama
label variable vitama "VITAMIN -A"

gen ofruits=1 if  v414l==1
label value ofruits ofruits
label variable ofruits "OTHER fruits"

egen dietindC = rsum(grains lagumes dairyp fleshf vitama ofruits eggs) if grains~=.| lagumes~=.|
dairyp~=.| fleshf~=.| vitama~=.| ofruits~=.| eggs~=.
label variable dietindC "Minimum dietary diversity children"


recode dietindC (1/3 .=0 "No") (4/7=1 "Yes"), gen(diet4)
ta diet4  [iw=v005/1000000]

*this gives 27.6% and 2442 children as in the report
```